

wscript(vbs) 控制OpenOffice

建立日期:2009/11/06

作者:cschen33.tw@gmail.com

目前大部份的程式(ERP)會呼叫office(word或是excel)來產生要的報表,因為免費軟體OpenOffice已經越來越穩定,是不是也可以使用 OpenOffice取代Office嗎?

目前在網路上找到一篇

http://wiki.services.openoffice.org/wiki/Documentation/DevGuide/ProUNO/Bridge/Automation_Bridge

http://www.kalitech.fr/clients/doc/VB_APIOOo_en.html

想要依樣畫葫蘆但是沒有scalac,後來也找到好幾篇,但是根據文章來實作,發現有很多問題都不可行,因此自己錄製OOo的巨集,並猜測試可行!

要求

支援的作業平台 (Windows 95, 98, NT4, ME, 2000, XP)

快速入門

1.calc

```
Set oSM =CreateObject("com.sun.star.ServiceManager")
'Set oCR = oSM.createInstance("com.sun.star.reflection.CoreReflection")
Set oDesktop = oSM.createInstance("com.sun.star.frame.Desktop")
Dim aArgs()

Set oDoc = oDesktop.loadComponentFromURL("private:factory/scalc","_blank",0,aArgs)

Set oSheet = oDoc.getSheets().getByIndex(0)

Call oSheet.getCellByPosition(0,0).SetValue(1000.32)
Call oSheet.getCellByPosition(0,1).SetString("abc")
```

2.writer

```
Dim oSM
Dim oDesk
Dim arg()

'Instantiate OOo : this line is mandatory with VB for OOo API
Set oSM = CreateObject("com.sun.star.ServiceManager")
'Create the first and most important service
Set oDesk = oSM.createInstance("com.sun.star.frame.Desktop")

'建立一個新的文件
Set oDoc = oDesk.loadComponentFromURL("private:factory/swriter", "_blank", 0, arg())
'關閉這個文件
oDoc.Close (True)
Set oDoc = Nothing

'開啟一個已經存在的檔案(請注意第一個參數)
Set oDoc = oDesk.loadComponentFromURL("file:///c:/dev/ooo/test.doc", "_blank", 0, arg())
'儲存檔案
Call oDoc.storeToURL("file:///c:/dev/ooo/test2.sxw", arg())
'關閉檔案
oDoc.Close (True)
Set oDoc = Nothing
```

將上述兩檔存成副檔名vbs,就可以直接執行,有人執行會出現dos的視窗,是因為它是使用cscript.exe來執行,你可

以改成wscript.exe來執行

對程式按滑鼠右鍵選開起檔案,選擇你要的軟體,若沒有可以選擇windows安裝路徑下的system32\wscript.exe

第一個範例:

excel 比較簡單,我們先說明之

'第一個定意新的物件com.sun.star.ServiceManager

```
Set oSM = CreateObject("com.sun.star.ServiceManager")
```

'建立一個實體

```
Set oDesktop = oSM.CreateInstance("com.sun.star.frame.Desktop")
```

'定義讀取參數,我們不會變,因此定義的空陣列,一般寫入才會變

```
Dim aArgs()
```

'設定要開啟新的檔案calc

```
Set oDoc = oDesktop.loadComponentFromURL("private:factory/scalc", "_blank", 0, aArgs)
```

'取得第一個空白頁

```
Set oSheet = oDoc.getSheets().getByIndex(0)
```

'輸入值,SetValue是設定數字,SetString是設定文字

```
Call oSheet.getCellByPosition(0,0).SetValue(1000.32)
```

```
Call oSheet.getCellByPosition(0,1).SetString("abc")
```

第二個範例:

'這裏有一個多了定義,其實也可以不用定義,在wscript 不需要定義as object否則會有錯誤訊息產生

```
Dim oSM 'Root object for accessing OpenOffice from VB
```

```
Dim oDesk, oDoc As Object 'First objects from the API
```

'定義讀取參數,我們不會變更,因此定義空陣列,一般寫入才會變

```
Dim arg()
```

'第一個定意新的物件com.sun.star.ServiceManager

```
Set oSM = CreateObject("com.sun.star.ServiceManager")
```

'建立一個實體

```
Set oDesk = oSM.CreateInstance("com.sun.star.frame.Desktop")
```

'開啟一個新的文件writer

```
Set oDoc = oDesk.loadComponentFromURL("private:factory/swriter", "_blank", 0, arg())
```

'關閉這個文件

```
oDoc.Close (True)
```

```
Set oDoc = Nothing
```

'開啟一個已經存在的檔案(pay attention to the syntax for first argument)

```
Set oDoc = oDesk.loadComponentFromURL("file:///c:/dev/ooo/test.doc", "_blank", 0, arg())
```

'儲存檔案,因為我們沒有要變更模式因此不變更寫入參數

```
Call oDoc.storeToURL("file:///c:/dev/ooo/test2.sxw", arg())
```

'關閉檔案

```
oDoc.Close (True)
```

'刪除物件

```
Set oDoc = Nothing
```

特殊函式

在OOo API 中有一些屬性的類型是structure,它是使用者自定的類型,並無法直接存取,要使用Bridge_GetStruct來取得屬性,及指定屬性

```
'  
'建立及設定com.sun.star.beans.PropertyValue 屬性  
'  
Public Function MakePropertyValue(cName, uValue)  
Dim oStruct, oServiceManager  
Set oServiceManager = CreateObject("com.sun.star.ServiceManager")  
Set oStruct = oServiceManager.Bridge_GetStruct("com.sun.star.beans.PropertyValue")  
oStruct.Name = cName  
oStruct.Value = uValue  
'設定傳回值  
Set MakePropertyValue = oStruct  
  
End Function
```

陣列

'定義,是必須的,vbs可以定義空的陣列,但是autoit3 不可以定義空的陣列

Dim stest(0)

'指定

Set stest(0) = MakePropertyValue("ToPoint", "\$A\$4")

'引用

stest()

#wscript的定義陣列 ,我試了好久,找一下資料才找到

Dim abc(3)

可以存4個值abc(0),abc(1),abc(2),abc(3)

Dim abc(0)

可以存1個值

輸入值的方法

1.

oSheet.getCellByPosition(0,0).SetValue 1000.32

2.

Call oSheet.getCellByPosition(2,2).SetValue(1000.32)

3.

Call oSheet.getCellByPosition(1,1).SetString("abc")

4.

oSheet.getCellByPosition(1,1).Setstring "abc"

5.

oSheet.getCellRangeByName("B1").Value = 12

1和2是相同的

3和4是相同的

選取

選取B3

call oDesktop.CurrentComponent.CurrentController.select(oSheet.getCellRangeByName("B3"))

選取B3:G5(B3到G5)

call oDesktop.CurrentComponent.CurrentController.select(oSheet.getCellRangeByName("B3:G5"))

'置中

Dim t3(0)

set t3(0) = MakePropertyValue("HorizontalAlignment" , 2)

call dispatcher.executeDispatch(document, ".uno:HorizontalAlignment", "", 0, t3)

'原OOo巨集中的com.sun.star.table.CellHoriJustify.CENTER的值是無法直接取得,因此使用數字來取代

'0正常,1靠左,2置中,3靠右

'移動

'移動到a3:f17,這是一個選取的空間

set t4(0) = MakePropertyValue("ToPoint", "\$A\$3:\$F\$17")

call dispatcher.executeDispatch(document, ".uno:GoToCell", "", 0, t4)

'開啟檔案

'路徑c:\tmp\sample.rtf 變更為file:///c:\tmp\sample.rtf,其中 |代表:

Set oDoc = oDesktop.loadComponentFromURL("file:///c:\tmp\sample.rtf", "_blank", 0, aArgs)

附件共3件

1.scalc.vbs

2.swriter.vbs

3.autoit3.vbs

```

1: ' 建立com
2: Set oSM = CreateObject("com.sun.star.ServiceManager")
3: ' 建立實體
4: Set oDesktop = oSM.CreateInstance("com.sun.star.frame.Desktop")
5: ' 定義陣列, vbs的陣列定義和一般不同: 如 t1(0), 可以存1個0->0 一個, 而t1(1), 可以存2個0->1
6: Dim aArgs(), t1(0), t2(7), t3(0), t4(0), SaveParam(0)
7: Set oDoc = oDesktop.loadComponentFromURL("private:factory/scalc", "_blank", 0, aArgs)
8: Set oSheet = oDoc.getSheets().getByIndex(0)
9: Set document = oDesktop.CurrentComponent.CurrentController.Frame
10: Set dispatcher = oSM.CreateInstance("com.sun.star.frame.DispatchHelper")
11:
12: ' 輸入數字
13: oSheet.getCellByPosition(0, 0).SetValue 1
14: Call oSheet.getCellByPosition(0, 1).SetValue(2)
15: oSheet.getCellRangeByName("A3").Value = 3
16: oSheet.getCellRangeByName("A4").SetValue 4
17:
18: ' 輸入字串
19: oSheet.getCellByPosition(1, 0).SetString "a"
20: Call oSheet.getCellByPosition(1, 1).SetString("b")
21:
22: ' 讀取值
23: oCell = oSheet.getCellRangeByName("b3").Value
24: MsgBox oCell
25:
26: ' 選取欄位並輸入文字
27: oSheet.getCellRangeByName("D8").SetString("中文字測試")
28:
29: ' 設字型大小
30: call oDesktop.CurrentComponent.CurrentController.select(oSheet.getCellRangeByName("B3:G5"))
31: set t1(0) = MakePropertyValue("FontHeight.Height", 14)
32: call dispatcher.executedDispatch( document , ".uno:FontHeight" , "" , 0 , t1)
33:
34: ' 設定邊界
35: set t2(0) = MakePropertyValue("OuterBorder.LeftBorder", Array(0, 0, 2, 0))
36: set t2(1) = MakePropertyValue("OuterBorder.LeftDistance", 0)
37: set t2(2) = MakePropertyValue("OuterBorder.RightBorder", Array(0, 0, 2, 0))
38: set t2(3) = MakePropertyValue("OuterBorder.RightDistance", 0)
39: set t2(4) = MakePropertyValue("OuterBorder.TopBorder", Array(0, 0, 2, 0))
40: set t2(5) = MakePropertyValue("OuterBorder.TopDistance", 0)
41: set t2(6) = MakePropertyValue("OuterBorder.BottomBorder", Array(0, 0, 2, 0))
42: set t2(7) = MakePropertyValue("OuterBorder.BottomDistance", 0)
43: call dispatcher.executedDispatch( document , ".uno:SetBorderStyle" , "" , 0 , t2)
44:
45: ' 移動
46: set t4(0) = MakePropertyValue("ToPoint", "$A$3:$F$17")
47: call dispatcher.executedDispatch(document, ".uno:GoToCell", "", 0, t4)
48:
49: ' 置中
50: ' com.sun.star.table.CellHoriJustify.CENTER的值是無法直接取得, 因此使用數字來取代
51: ' 0正常, 1靠左, 2置中, 3靠右
52: set t3(0) = MakePropertyValue("HorizontalAlignment" , 2)
53: call dispatcher.executedDispatch(document, ".uno:HorizontalAlignment" , "" , 0 , t3)
54:
55:
56: ' 存成PDF
57: Set SaveParam(0) = MakePropertyValue("FilterName", "writer_pdf_Export")
58: Call oDoc.storeToURL("file:///C:/tmp/testdoc.pdf", SaveParam)
59:
60: Function MakePropertyValue(cName, uValue)
61:     Dim oPropertyValue
62:     Dim oSM
63:     Set oSM = CreateObject("com.sun.star.ServiceManager")
64:     Set oPropertyValue = oSM.Bridge_GetStruct("com.sun.star.beans.PropertyValue")
65:     oPropertyValue.Name = cName
66:     oPropertyValue.Value = uValue
67:     Set MakePropertyValue = oPropertyValue
68:
69: End Function
70:
71:

```

```
1: ' VBScript 範例
2: ' The service manager is always the starting point
3: ' 定義 service manager
4: Set objServiceManager= WScript.CreateObject("com.sun.star.ServiceManager")
5:
6: ' 建立實體
7: Set objCoreReflection= objServiceManager.createInstance("com.sun.star.reflection.CoreReflection")
8:
9: ' 建立在桌面
10: Set objDesktop= objServiceManager.createInstance("com.sun.star.frame.Desktop")
11:
12: ' 開啟新的文件
13: Dim args()
14: Set objDocument= objDesktop.loadComponentFromURL("private:factory/swriter", "_blank", 0, args)
15:
16: ' 定義文字物件
17: Set objText= objDocument.getText
18:
19: ' 建立一個暫時的物件
20: Set objCursor= objText.createTextCursor
21:
22: ' 插入一些文字
23: objText.insertString objCursor, "第一行在新建的文件中." & vbLf, false
24:
25: ' 再插入第二行
26: objText.insertString objCursor, "Now we're in the second line", false
27:
28: ' 建立一個文字表格4*4
29: Set objTable= objDocument.createInstance("com.sun.star.text.TextTable")
30: objTable.initialize 4, 4
31:
32: ' 插入表格
33: objText.insertTextContent objCursor, objTable, false
34:
35: ' 取得一行
36: Set objRows= objTable.getRows
37: Set objRow= objRows.getByIndex( 0)
38:
39: ' 設定表格的背景顏色
40: objTable.setPropertyValue "BackTransparent", false
41: objTable.setPropertyValue "BackColor", 13421823
42:
43: ' 設定第一列使用不同的背景顏色
44: objRow.setPropertyValue "BackTransparent", false
45: objRow.setPropertyValue "BackColor", 6710932
46:
47: ' 填入表格第一列
48: insertIntoCell "A1", "FirstColumn", objTable
49: ' insertIntoCell is a helper function, see below
50: insertIntoCell "B1", "SecondColumn", objTable
51: insertIntoCell "C1", "ThirdColumn", objTable
52: insertIntoCell "D1", "SUM", objTable
53:
54: objTable.getCellByName("A2").setValue 22.5
55: objTable.getCellByName("B2").setValue 5615.3
56: objTable.getCellByName("C2").setValue -2315.7
57: objTable.getCellByName("D2").setFormula "sum "
58:
59: objTable.getCellByName("A3").setValue 21.5
60: objTable.getCellByName("B3").setValue 615.3
61: objTable.getCellByName("C3").setValue -315.7
62: objTable.getCellByName("D3").setFormula "sum "
63:
64: objTable.getCellByName("A4").setValue 121.5
65: objTable.getCellByName("B4").setValue -615.3
66: objTable.getCellByName("C4").setValue 415.7
67: objTable.getCellByName("D4").setFormula "sum "
68:
69: ' Change the CharColor and add a Shadow
70: objCursor.setPropertyValue "CharColor", 255
71: objCursor.setPropertyValue "CharShadowed", true
72:
73: ' Create a paragraph break
```

```
74: ' The second argument is a com::sun::star::text::ControlCharacter::PARAGRAPH_BREAK constant
75: objText.insertControlCharacter objCursor, 0 , false
76:
77: ' Inserting colored Text.
78: objText.insertString objCursor, " This is a colored Text - blue with shadow" & vbLf, false
79:
80: ' Create a paragraph break ( ControlCharacter::PARAGRAPH_BREAK).
81: objText.insertControlCharacter objCursor, 0, false
82:
83: ' Create a TextFrame.
84: Set objTextFrame= objDocument.CreateInstance("com.sun.star.text.TextFrame")
85:
86: ' Create a Size struct.
87: Set objSize= createStruct("com.sun.star.awt.Size")
88: ' helper function, see below
89: objSize.Width= 15000
90: objSize.Height= 400
91: objTextFrame.setSize( objSize)
92:
93: ' TextContentAnchorType.AS_CHARACTER = 1
94: objTextFrame.setPropertyValue "AnchorType", 1
95:
96: ' insert the frame
97: objText.insertTextContent objCursor, objTextFrame, false
98:
99: ' Get the text object of the frame
100: Set objFrameText= objTextFrame.getText
101:
102: ' Create a cursor object
103: Set objFrameTextCursor= objFrameText.createTextCursor
104:
105: ' Inserting some Text
106: objFrameText.insertString objFrameTextCursor, "The first line in the newly created text frame.", _
107: false
108: objFrameText.insertString objFrameTextCursor, _
109: vbLf & "With this second line the height of the frame raises.", false
110:
111: ' Create a paragraph break
112: ' The second argument is a com::sun::star::text::ControlCharacter::PARAGRAPH_BREAK constant
113: objFrameText.insertControlCharacter objCursor, 0 , false
114:
115: ' Change the CharColor and add a Shadow
116: objCursor.setPropertyValue "CharColor", 65536
117: objCursor.setPropertyValue "CharShadowed", false
118:
119: ' Insert another string
120: objText.insertString objCursor, " That's all for now !!", false
121:
122: On Error Resume Next
123:     If Err Then
124:         MsgBox "An error occurred"
125:     End If
126:
127:
128: Sub insertIntoCell( strCellName, strText, objTable)
129:     Set objCellText= objTable.getCellByName( strCellName)
130:     Set objCellCursor= objCellText.createTextCursor
131:     objCellCursor.setPropertyValue "CharColor", 16777215
132:     objCellText.insertString objCellCursor, strText, false
133: End Sub
134:
135: Function createStruct( strTypeName)
136:     Set classSize= objCoreReflection.forName( strTypeName)
137:     Dim aStruct
138:     classSize.createObject aStruct
139:     Set createStruct= aStruct
140: End Function
141:
142:
```

```
1: #include <Array.au3>
2: Global $set00oProp
3: $oSM=ObjCreate("com.sun.star.ServiceManager")
4: if IsObj($oSM) Then
5:     $oDesktop = $oSM.CreateInstance("com.sun.star.frame.Desktop")
6:
7:     If IsObj($oDesktop) Then
8:         Dim $aArgs[1]
9:         Dim $t1[1]
10:        Dim $t2[1]
11:        $aArgs[0] = setProp("ReadOnly",False)
12:        ;$aArgs[1] = setProp("Password","secret")
13:        ;$aArgs[2] = setProp("Hidden",False)
14:        $oDoc = $oDesktop.loadComponentFromURL("private:factory/scalc","_blank",0,$aArgs)
15:        $document = $oDesktop.CurrentComponent.CurrentController.Frame
16:        $dispatcher = $oSM.CreateInstance("com.sun.star.frame.DispatchHelper")
17:        If IsObj($oDoc) Then
18:            $oSheet = $oDoc.GetSheets().getByIndex(0)
19:            $oSheet.getCellByPosition(5,5).SetString("中文字測試")
20:            $oDesktop.CurrentComponent.CurrentController.select($oSheet.getCellByPosition(5,5))
21:            $t2[0] = setProp("FontHeight.Height",14)
22:            $dispatcher.executedDispatch($document,".uno:FontHeight","",0,$t2)
23:            $t1[0] = setProp("ToPoint","A3:F17")
24:            $dispatcher.executedDispatch($document,".uno:GoToCell","",0,$t1)
25:        Else
26:            MsgBox(0,"oDoc","not object")
27:        EndIf
28:    Else
29:        MsgBox(0,"oDesktop","not object")
30:    EndIf
31: Else
32:    MsgBox(0,"oExcel","not object")
33: EndIf
34:
35: Func setProp($cName , $uValue)
36:     $oSM = ObjCreate("com.sun.star.ServiceManager")
37:     $oPropertyValue = $oSM.Bridge_GetStruct("com.sun.star.beans.PropertyValue")
38:     $oPropertyValue.Name = $cName
39:     $oPropertyValue.Value = $uValue
40:     $set00oProp = $oPropertyValue
41:     Return $set00oProp
42: EndFunc
43:
```